

Customize Your Table of Contents with the ODS Destination for Word

David W. Kelley, SAS Institute Inc., Cary, NC

ABSTRACT

The Base SAS® Output Delivery System (ODS) destination for Word enables customers to deliver SAS® reports as native Microsoft Word documents. ODS WORD generates reports in the Office Open XML Document (.docx) format, which has been standard in Microsoft Word since 2007. The .docx format uses ZIP compression, which makes for a smaller storage footprint and speedier downloading. ODS WORD is preproduction in the sixth maintenance release of SAS 9.4.

This paper shows you how to make a custom table of contents (TOC) with ODS WORD. You will learn how to control the placement, text, and style of your TOC. Place your TOC anywhere in the body of your document. Make your TOC title and entry text anything you want it to be. Assign custom colors or fonts to your TOC text. Even put a stylish border around your TOC if you like!

Adding a TOC to your document makes it more navigable, whether it is in digital or hard-copy format. Adding a custom TOC makes your document smarter, which makes you look smarter too! Whatever your SAS programming experience level may be, you will benefit from this session.

INTRODUCTION

The ODS TOC contextualizes the tables, text, and graphs (output objects) that comprise the body of a report generated by a destination such as ODS WORD. It does this by imposing an internally defined hierarchical structure on the output objects. The TOC also aids navigation to the individual output objects. When viewing the report, clicking an entry in the TOC causes the viewing application to reposition to the associated output object.

ODS generates the TOC automatically. ODS and SAS procedures cooperate to make this happen. Procedures provide ODS the hierarchical groupings for the output objects that they create. ODS defines additional hierarchy based on the context, such as when BY grouping is in effect. All you do is request the TOC, and ODS takes care of the rest.

In most cases, the TOC that ODS makes for you "out of the box" is not ready to be consumed by your customers. You need to customize it somehow, so that it meets their requirements. One requirement, no doubt, is that the TOC's content not be obviously machine-generated. You win if your TOC looks handcrafted even though it is an artifact of automation.

ODS and reporting procedures provide options for customizing the content of the TOC. These options behave the same way across destinations. What works for ODS HTML5 also works for ODS PDF, ODS RTF, and ODS WORD.

Customizing the appearance of the TOC is a different story. For any given destination, you use PROC TEMPLATE to define a custom style that will govern fonts, colors, and other visual aspects of your report's elements, including the TOC. But how that style is defined is

specific to the destination. What works one way in one destination may work another way, or not at all, in another destination. This is especially true of the TOC in ODS WORD. The elements of the ODS WORD default style (Styles.Word) that control the appearance of the TOC have little or no effect in other destinations.

This means that what you will learn to customize the ODS WORD TOC is truly custom. The techniques that we consider in this paper are specialized, and in some cases, advanced. This paper aims for depth. If you prefer breadth, then consider reading my SAS Global Forum 2019 paper ([Kelley, 2019](#)) first, and then rejoining this discussion. Let's get started!

MY TUNES

Data is not the primary focus of this paper, but we need some in order to create ODS WORD reports. For convenience's sake, I'm revisiting my iTunes music playlist, which I initially described in my SAS Global Forum 2019 paper ([Kelley, 2019](#)). I updated the playlist by synchronizing my iPod with the iTunes app on my PC, then exporting the playlist as a text file. Running the IMPORT procedure and the DATA step on the playlist text file yielded the SAS data set called WORK.MYITUNES. Table 1 shows the variables for WORK.MYITUNES.

Variable	Type	Format
Album	Char	\$95.
Artist	Char	\$255.
Composer	Char	\$118.
Date_Added	Num	DATETIME.
Genre	Char	\$18.
Last_Played	Num	DATETIME.
Last_Skipped	Num	DATETIME.
Name	Char	\$456.
Plays	Num	BEST12.
Skips	Num	BEST12.
Time	Num	MMSS.
Total_Time	Num	HHMM.
Track_Count	Num	BEST12.
Track_Number	Num	BEST12.
Year	Char	\$4.

Table 1: WORK.MYITUNES Variables

WORK.MYITUNES represents the tracks on my iPod. The Plays variable records how many times a track has been played. The Time variable records the duration of the track. The Total_Time variable is the product of Plays and Time; it represents the total amount of time a track has been played. Total time strikes me as a reasonable measure of how much I like a track. Over time, I'd spend more time playing more favored tracks than less favored tracks, don't you think?

That said, I tend to listen to favorite artists more so than favorite tracks. Typically, I don't jump around from track to track regardless of the artist. That's because I'm often listening while commuting to or from work, and fiddling with the iTunes player is not a driving best practice! (And I'm not a fan of the shuffle feature.) For the purposes of this discussion, I'm interested in reporting on the total time spent playing tracks for a particular artist. And for that we'll use a Top N report.

TOPS OF THE POPS

Perhaps it's more accurate to say that we'll *reuse* a Top N report, seeing how it's a reporting technique previously described in my SAS Global Forum paper ([Kelley, 2019](#)). Top N reports are good for ranking items by one or more measures. Here we want to rank the Top N artists in total playing time by genre. The main steps of the algorithm to create the report are as follows:

1. Summarize the WORK.MYITUNES total playing time by genre and artist, and then store the results in output data set WORK.SUMMARY.
2. Sort WORK.SUMMARY so that the top total times by genre appear at the top of output data set WORK.TOPN.
3. Rank the WORK.TOPN top 10 or fewer total times by genre.
4. Print the WORK.TOPN top 10 or fewer artists in total time by genre.

Because the TOC is the primary focus of this paper, we instruct ODS WORD to generate a TOC for the report.

Here is the code for the Top N report:

```
proc means data=myitunes sum noprint;
  var total_time;
  class genre artist;
  output out=summary sum=total_time genre /levels;
run;

proc sort data=summary out=topn;
  where _type_>2;
  by genre descending total_time;
run;

data topn;
  length rank 8;
  label rank="Rank";
  set topn;
  by genre descending total_time;
  if nmiss(of total_time) then delete;
  if first.genre then rank=0;
  rank+1;
  if rank le 10 then output;
run;

ods word file="c:\users\sasdck\onedrive - sas\topN1.docx"
  options(contents="on" toc_data="on" keep_next="on");
title "Top N Artists in Total Playing Time by Genre";
title2 "Time format is (Hours:Minutes).";
proc print data=topn noobs label;
```


```
by genre;  
var rank artist total_time;  
run;  
ods word close;
```

Like many destinations, ODS WORD supports an OPTIONS argument for controlling behaviors that are specific to the destination. Specifying "on" for the CONTENTS= option tells ODS WORD to generate a TOC. TOC_DATA="on" tells ODS WORD to generate TOC entries. You might think that making a TOC implies making TOC entries, and thus having separate options for the two operations is redundant. However, we'll see later why this arrangement is necessary.

KEEP_NEXT="on" gives Microsoft Word a hint to keep the BY line (e.g. "Genre=Blues") on the same page as the table immediately following it. You don't want Word to render the BY line at the bottom of one page and its associated table at the top of the next page. (In typesetting, such text is referred to as an *orphan*.)

Note that Word doesn't automatically display the TOC when you open the document. You can manually display the TOC by pressing CTRL+A F9 on your keyboard. Another Word tip: If you jump from the TOC to the body of the report, then you can jump back to the TOC by pressing ALT ←.

Display 1 shows how Word renders the TOC for the Top N report. This is what you get by default, i.e. with no customization of the TOC's content or style. Frankly, it's not much to look at. Clearly, we can do better, and we will!




Display 1: Default TOC for Top N Report

LESS IS MORE

Out of the box, the TOC for the Top N report is cluttered with unhelpful entries. The first-level (outermost) entry has the label "The Print Procedure", which is meaningful to you, the SAS developer, but not to your customer. The third-level (innermost) entries are both meaningless and redundant. The first order of business is to modify the TOC to display only the second-level (middle) entries.

While we're modifying the TOC's content, let's also modify its style. Specifically, let's change the font color and style of the entries. Display 2 shows version 2 of the TOC.



Display 2: TOC with Customized Entries

That's better! Now the TOC has only useful entries.

Notice that Display 2 shows only the TOC, and not the body of the report. That's because the body is unchanged from before, and the TOC is what this paper is about, anyway.

Here is the code for version 2 of the Top N report, sans the data preparation steps:

```
ods path (prepend) work.templat(update);
proc template;
  define style styles.mystyle;
    parent = styles.Word;
    class toc2 /
      color=very dark red
      fontstyle=italic
    ;
  end;
run;

ods word file="c:\users\sasdck\onedrive - sas\topN2.docx" style=mystyle
  options(contents="on" toc_data="on" keep_next="on");
title "Top N Artists in Total Playing Time by Genre";
title2 "Time format is (Hours:Minutes).";
ods proclabel=" ";
```

```
proc print data=topn noobs label contents="";
  by genre;
  var rank artist total_time;
run;
ods word close;
```

The ODS PROCLABEL statement controls the content of the first-level TOC entry. Setting the label to " " removes the entry from the TOC. The PROC PRINT CONTENTS option controls the content of the third-level TOC entries. Setting the text to "" removes the entries from the TOC. That leaves only the second-level entries to be displayed.

The PROC TEMPLATE code defines a custom style called Styles.MyStyle. Styles.MyStyle inherits from the ODS WORD default style, Styles.Word. This means that Styles.MyStyle will have the same fonts, colors, etc. as Styles.Word, except that Styles.MyStyle overrides them. (And in case you're wondering, style names are case-insensitive.)

Microsoft Word controls the appearance of TOC entries by level. It supports up to nine nested levels of styling. In PROC TEMPLATE, for Styles.Word or styles inheriting from Styles.Word, those nine levels are represented by the style elements TOC1, TOC2, ..., TOC9.


For version 2 of the TOC, we want to modify the appearance of the second-level entries, which are represented by TOC2 in PROC TEMPLATE. We change the TOC2 font color to be very dark red and the style to be italic. That's all there is to it!

You may wonder why the style element is TOC2 and not TOC1. Version 2 of the TOC has only one level of entries, after all. The style is defined that way so that you don't necessarily have to recode it if the TOC content changes.

You can customize more of the TOC than just its entries. The next example shows you how.

TOC OF THE TOWN

You observed in the previous example that we customized the TOC entries using a PROC TEMPLATE style element. We can use a similar technique to customize the TOC title and borders of the Top N report. Display 3 shows version 3 of the TOC.



Display 3: TOC with Customized Entries, Title, and Border

For version 3, we change the content and style of the title of the TOC. We also add a border around the TOC.

Here is the code for version 3 of the Top N report:

```
ods path (prepend) work.templat(update);
proc template;
  define style styles.mystyle;
    parent = styles.word;
    class contenttitle /
      content = "Top N Artists in Total Playing Time by Genre"
      fontweight = bold
      color=dark blue
      margintop=12pt
    ;
    class toc2 /
      color=very dark red
      fontstyle=italic
    ;
    class contents /
      borderstyle=double
      bordercolor=green
```



```

        ;
    end;
run;

ods word file="c:\users\sasdck\onedrive - sas\topN4.docx" style=mystyle
    options(contents="on" toc_data="on" keep_next="on");
title "Top N Artists in Total Playing Time by Genre";
title2 "Time format is (Hours:Minutes).";
ods proclabel=" ";
proc print data=topn noobs contents="" label;
    by genre;
    var rank artist total_time;
run;
ods word close;

```

To modify the title of the TOC, in PROC TEMPLATE we edit the CONTENTTITLE style element. To change the TOC title content, we set the CONTENT style attribute to the desired text. (And, yes, it's confusing and unfamiliar to talk about the CONTENT attribute of the CONTENTTITLE element. Bear with me.) To change the TOC title appearance, we set the familiar font weight and color attributes. Additionally, we define the top margin such that there will be a visually pleasing amount of vertical space between the top border and the title.

To instruct ODS WORD to draw a border around the TOC, we customize the CONTENTS style element. There are multiple ways to define a border:

1. Specify a border style using the BORDERSTYLE attribute. You can choose from DASHED, DOTTED, DOUBLE, INSET, GROOVE, OUTSET, RIDGE, and SOLID. ODS WORD will draw a border of the specified style using the default border width.
2. Specify a border width using the BORDERWIDTH attribute. ODS WORD will draw a border of the specified width using the default border style.
3. Specify a border style and width. This should be self-explanatory.

Besides the style and width, you can specify the color of the border. Version 3 of the TOC sports a green double border.

It's not shown in this example, but you can set the style, width, and color separately for the top, bottom, left, and right borders of the TOC. Just don't get carried away!

Also, not shown here is the ability to place the TOC at the end of the Word document instead of at the beginning. You can do this by specifying the VJUST attribute of the CONTENTS style element. Setting VJUST to B (bottom) will place the TOC at the end of your report. By default, VJUST is T (top).

Careful readers may have noticed that the third code example generates a file named topN4.docx and were expecting it to be named topN3.docx instead. They're not wrong. Originally, I included an example that demonstrates a report with the TOC at the end, but later I removed it in order to keep this paper at a manageable length. Not to worry! You can find all the example code, including the deleted example, at my SAS Global Forum 2020 site ([Kelley, 2020](#)).

DIY TOC

Remember earlier when I said that you win if your customer can't tell that your ODS WORD report is machine-generated? Well, with version 3 of the Top N report, I judge that we're almost but not quite winning. A couple of things still stand out like a sore thumb, namely:

1. Each TOC entry is prefixed with "Genre=", which is traditional SAS BY line behavior. Of course, the BY lines also appear in the body of the report. Can we replace the BY line behavior with something more natural?
2. The body of the report is comprised of tables, but what the report is describing is *lists*. Why not use lists, then?

Version 4 of the Top N report is intended to address these shortcomings. Display 4 shows what I have in mind.

The screenshot shows a Microsoft Word document titled "topN5.docx - Saved to OneDrive - SAS" with the "View" ribbon selected. The document is split into two panes. The left pane displays a Table of Contents (TOC) for "Top N Artists in Total Playing Time by Genre". The right pane displays the report body, which is a list of artists and their playing times, grouped by genre. The report body includes a date stamp "Monday, January 20, 2020 02:13:36 PM" and a time format note "Time format is (Hours:Minutes)".

Genre	Artist	Playing Time
Alternative Rock	Old Crow Medicine Show	0:04
Alternative Rock	Lucinda Williams	0:03
Blues	B.B. King	4:42
Blues	B.B. King & Bobby Bland	0:50
Blues	Little Walter	0:09
Blues	Sonny Boy Williamson	0:07
Blues	Hawtin' Wolf	0:06
Blues	John Lee Hooker	0:06
Blues	Buddy Guy	0:06
Blues	Koko Taylor	0:06
Blues	Willie Dixon	0:06
Blues	Jimmy Rogers	0:06
Children's	Disney	0:59
Children's	Julie Andrews	0:35
Children's	Amy Lou Barnes, Sally Mueller, Mary Moder, Alice Sizer, Betty Bayne	0:23
Children's	David Tomlinson/Dick Van Dyke/The Londoners	0:15
Children's	Cliff Edwards, Jim Carmichael & The Hall Johnson Choir	0:06
Children's	Dickie Jones	0:05
Children's	Peggy Lee	0:04
Children's	Peggy Lee & The Mello Men	0:03
Children's	Samuel E. Wright	0:03
Children's	Burl Ives	0:02
Classical	All-Time Greatest Hits	7:28
Country	Johnny Cash	5:35
Country & Folk	Bob Dylan	21:54
Country & Folk	Willie Nelson	15:47
Country & Folk	Gordon Lightfoot	8:28
Country & Folk	Doc Watson	7:59
Country & Folk	Johnny Cash	6:29
Country & Folk	Emmylou Harris	3:40

Display 4: TOC Entries without "Genre=" Prefix

As you can see, the revised report looks quite different. The TOC and body of the report are more concise and readable. How did I do that? Here is the code:

```
ods path (prepend) work.templat(update);
proc template;
  define style styles.mystyle;
    parent = styles.word;
    class contenttitle /
      content = "Top N Artists in Total Playing Time by Genre"
      fontweight = bold
      color=dark blue
      margintop=12pt
    ;
    class toc2 /
      color=very dark red
      fontstyle=italic
    ;
    class contents /
      borderstyle=double
      bordercolor=green
    ;
  end;
run;

ods word file="c:\users\sasdck\onedrive - sas\topN5.docx" style=mystyle
  options(contents="on" toc_data="on" toc_type="headings");
title "Top N Artists in Total Playing Time by Genre";
title2 "Time format is (Hours:Minutes).";

%macro doOneGenre(val);
proc odstext;
  h2 "&val"; /*-- h2 matches up with toc2. --*/
run;
proc odslist data=topn(where=(genre="&val"));
  item strip(put(artist,$255.)) || " (" || strip(put(total_time,hhmm.)) ||
  ")"/style={liststyletype=decimal};
run;
%mend;

%doOneGenre(Alternative Rock);
%doOneGenre(Blues);
%doOneGenre(Children%STR('%')s);
%doOneGenre(Classical);
%doOneGenre(Country);
%doOneGenre(Country & Folk);
%doOneGenre(Folk);
%doOneGenre(Folk-Rock);
%doOneGenre(Jazz);
%doOneGenre(Pop);
%doOneGenre(%NRSTR(R&B));
%doOneGenre(Religious);
%doOneGenre(Rock);
%doOneGenre(Soundtrack);
%doOneGenre(Traditional);
```

```
ods word close;
```

Before we dive into what has changed in the code, I'll point out what *hasn't* changed: the style definition. Version 4 of the Top N report addresses content, not style.

The first important change to the Top N report is to replace the BY lines. We want to lose the "Genre=" prefix in each TOC entry, as well as in each report body BY line. We need to take control over the Genre variable text. We do that with PROC ODSTEXT.

The PROC ODSTEXT H2 statement instructs ODS WORD to generate a second-level heading with the given text in the report body. We want that heading text also to serve as a TOC entry. Specifying TOC_TYPE="HEADINGS" in the ODS WORD OPTIONS argument accomplishes that for us.

Basically, we're telling ODS WORD that the report writer, not ODS, is responsible for structuring the TOC. In other words, this is a "Do it yourself" (DIY) TOC. A feature of the DIY TOC is that its structure is reflected in the report body, in the form of headings and subheadings. This is a customary way of organizing document content, and Microsoft Word supports it in a natural fashion.

You may wonder why I specified a second-level heading. That's because style Styles.Mystyle expects second-level TOC entries. If we use the PROC ODSTEXT H2 statement, then we can reuse the style as-is.

The second important change to the report is to replace tables with lists. We substitute PROC ODSLST for PROC PRINT. PROC ODSLST lists the specified contents of the input data set. In this case, we're listing the artists in order of decreasing total playing time. We filter the artists by genre.

The PROC ODSLST ITEM statement accepts an expression, which is evaluated for each observation in the input data set. We know from [Table 1](#) to specify a format of \$255. to the PUT function for variable Artist, and to specify a format of HHMM. to the PUT function for variable Total_Time. By default, lists are unordered (bulleted). I want an ordered list, so I specify LISTSTYLETYPE=DECIMAL as a style attribute override.

The example code builds the report using a macro. The code invokes the macro for each genre. The macro calls are hand-coded, which isn't a robust programming technique. I'll leave it to you to determine how to make the macro calls data-driven.

A PEEK UNDER THE HOOD

Microsoft Word provides the TOC functionality by way of a [field code](#). Field codes are text instructions that are evaluated when the Word document is opened, and they automatically update when necessary. The most common field codes are the date and page number. If you use Microsoft Word, then you've likely observed the page number of a longish document updating just after you open it.

The field code for the TOC is called ... *drum roll* ... TOC. The syntax of the TOC instruction looks like: {TOC [Switches]}.

Switches are options, of which there are many. We'll focus on just a few:

\f EntryIdentifier

Builds a table from TC fields that match *EntryIdentifier*.

\h

Inserts TOC entries as hyperlinks. Clicking an entry causes Word to reposition to the associated output object in the report body. This is the navigation behavior that we want.

\p "Separator"

Specifies the character that separates an entry and its page number. The default is a tab with leader dots.

\z

Hides the tab leader and page numbers in Web layout view. Web pages aren't paginated, so suppressing the leader and page numbers makes sense.

When you specify CONTENTS="ON" in the OPTIONS argument, ODS WORD scripts a TOC instruction that looks like this: {TOC \f C \h \z}. This tells Microsoft Word to build a table from TC fields that match C.

TC is the field code for a TOC entry. An example of a TC instruction is:

```
{TC "Alternative Rock" \f C \l 2}
```

This instruction defines a second-level TOC entry with the label "Alternative Rock".


When you specify TOC_DATA="ON" in the OPTIONS argument, ODS WORD scripts TC instructions in one of two ways:

1. For every grouping and output object it receives, by default.
2. For every heading it receives, if TOC_TYPE="HEADINGS".

At this point, you may be wondering why I'm belaboring the discussion with these ODS WORD implementation details. What's the payoff?

Well, ODS WORD provides a way for you to insert field instructions directly into the body of your report. This enables you to support TOC behaviors that aren't exposed by ODS WORD options. This is taking your DIY TOC to a whole new level!

Imagine that for version 5 of the Top N report, we want to dispense with the leader dots in the TOC entries. Display 5 shows the desired report:



Display 5: TOC without Leader Dots

Here is the SAS code that generates the report:

```
ods path (prepend) work.templat(update);
proc template;
  define style styles.mystyle;
    parent = styles.word;
    class contenttitle /
      content = "Top N Artists in Total Playing Time by Genre"
      fontweight = bold
      color=dark blue
      margintop=12pt
    ;
    class toc2 /
      color=very dark red
      fontstyle=italic
```

```

    ;
    class contents /
      borderstyle=double
      bordercolor=green
    ;
  end;
run;

options nodate nonumber;

ods escapechar='^';
ods word file="c:\users\sasdck\onedrive - sas\topN6.docx" style=mystyle
  options(toc_data="on" toc_type="headings");

proc odstext;
  p "Top N Artists in Total Playing Time by Genre"/style=contenttitle{just=c};
  p "^{run {TOC \f C \h \z \p " " "}}";
run;

title "Top N Artists in Total Playing Time by Genre";
title2 "Time format is (Hours:Minutes).";

options date number;

proc odstext pagebreak=yes;
  p ""/style={fontsize=.5pt}; /*-- 1/144th of an inch --*/
run;

%macro doOneGenre(val);
proc odstext;
  h2 "&val"; /*-- h2 matches up with toc2. --*/
run;
proc odslist data=topn(where=(genre="&val"));
  item strip(put(artist,$255.)) || " (" || strip(put(total_time,hhmm.))
  || ")"/style={liststyletype=decimal};
run;
%mend;

%doOneGenre(Alternative Rock);
%doOneGenre(Blues);
%doOneGenre(Children%STR(%)s);
%doOneGenre(Classical);
%doOneGenre(Country);
%doOneGenre(Country & Folk);
%doOneGenre(Folk);
%doOneGenre(Folk-Rock);
%doOneGenre(Jazz);
%doOneGenre(Pop);
%doOneGenre(%NRSTR(R&B));
%doOneGenre(Religious);
%doOneGenre(Rock);
%doOneGenre(Soundtrack);
%doOneGenre(Traditional);

```

```
ods word close;
```

Overall, the code for version 5 of the Top N report is similar to that for version 4. The modifications, in execution order, are:

1. Dates and page numbers are turned off. Typically, you don't want them on your TOC page.
2. The ODS escape character is set. This is an indication that we'll be doing inline processing.
3. The ODS WORD CONTENTS option is off, which is the default behavior. ODS WORD is not generating a TOC; we're doing it ourselves.
4. A new PROC ODSTEXT step:
 - o Specifies the TOC title.
 - o Uses the RUN inline command to specify the TOC instruction. The TOC instruction uses the \p option to specify that the TOC entry and its page number are to be separated by a single space character.
5. Dates and page numbers are turned on. Typically, you want them in the body of your report.
6. A new PROC ODSTEXT step inserts a page break, so that the Top N lists will begin on a separate page from the TOC. A side effect of this step is that a negligible amount of vertical space is inserted in the document.

Modification #4 is pivotal, and the RUN inline command is the pivot. The RUN command tells ODS WORD to insert text, which may include field instructions, at the current position in the report body. ODS WORD is oblivious to the content of the text.

Notice that Display 5 shows no border around the TOC. Style Styles.Mystyle specifies a green double border. Why is it not drawn? That's because ODS WORD doesn't know that we've requested a TOC, so it doesn't know to apply the CONTENTS style element definition to it. For the same reason, we had to explicitly specify the CONTENTTITLE style element when defining the TOC title.

By way of comparison, the TOC entries *do* have the style attributes set forth in the Styles.Mystyle TOC2 style element definition. That has to do with the way Microsoft Word built-in TOC styles are defined and is a happy accident from our perspective. The larger point is that if you stray from the confines of ODS WORD's built-in support for the TOC, then you may acquire desired new behaviors, but lose desired predefined behaviors. Fair warning!

CONCLUSION

A concise, informative TOC adds helpful context to your report, plus it makes it more navigable. ODS WORD provides the ability for you to create a TOC for your SAS report that is polished and looks handcrafted, even though it is a product of automation. Your customers will be none the wiser!

This paper supports my claim. I wrote it with ODS WORD. It's generated by a SAS program. Could you tell? If so, [contact me](#) and let me know how you knew, so that I can cover my tracks better in the future!

REFERENCES

- Kelley, David. 2019. "An Introduction to the ODS Destination for Word." *Proceedings of the SAS Global Forum 2019 Conference*. Cary, NC: SAS Institute Inc. Available at <https://www.sas.com/content/dam/SAS/support/en/sas-global-forum-proceedings/2019/3235-2019.pdf><https://www.sas.com/content/dam/SAS/support/en/sas-global-forum-proceedings/2019/3235-2019.pdf>.
- "List of field codes in Word." *Microsoft*. Available at <https://support.office.com/en-us/article/list-of-field-codes-in-word-1ad6d91a-55a7-4a8d-b535-cf7888659a51><https://support.office.com/en-us/article/list-of-field-codes-in-word-1ad6d91a-55a7-4a8d-b535-cf7888659a51>.
- Kelley, David. 2020. "Customize Your Table of Contents with the ODS Destination for Word." *SAS Institute Inc*. Available at <https://github.com/sascommunities/sas-global-forum-2020/tree/master/papers/4265-2020-Kelley><https://github.com/sascommunities/sas-global-forum-2020/tree/master/papers/4265-2020-Kelley>.

ACKNOWLEDGMENTS

The author would like to thank coworkers Jody Carlton, Allison Crutchfield, Bari Lawhorn, Kathryn McLawhorn, and Chevell Parker for reviewing this paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

David W. Kelley
SAS Institute Inc.
100 SAS Campus Drive
Cary, NC 27513-2414
David.Kelley@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.